



PSPICE tutorial

Present: Ngô Lam Trung

Dept. of Computer Engineering

Nội dung trình bày

- Giới thiệu tổng quan về Pspice
- Mô tả mạch bằng Pspice
- Các chế độ phân tích
- Ví dụ

Tổng quan về PSpice

- Để thực hiện kiểm tra hoạt động của mạch có 2 cách:
 - Chế tạo mạch thực sự theo thiết kế và kiểm tra hoạt động của mạch trên thực tế.
 - Chính xác, trực quan.
 - Tốn thời gian và chi phí chế tạo.
 - Phân tích và mô phỏng trên máy tính
 - Sử dụng ngôn ngữ mô tả mạch và công cụ mô phỏng trên máy tính.

Tổng quan về PSpice

- PSPICE là chương trình mô phỏng mạch kế thừa từ SPICE2, phát triển tại Berkeley

Simulation Program with Intergrated Circuit Emphasis

- Một số phiên bản khác của SPICE
 - HSPICE (Meta Software) và I-SPICE của NCSS: dùng cho mainframe.
 - AccuSim (MentorGraphics)
 - ViewSpice của ViewLogicSystem...
- PSpice do MicroSim (Cadence) phát triển

Ứng dụng của PSpice


- Kiểm tra hoạt động của thiết kế mạch trước khi sản xuất.
- Xác định kết quả hoạt động của mạch ở trạng thái lý tưởng.
- Thực hiện các phép đo lường khó thao tác trên thực tế: nhiễu, tín hiệu nhỏ, nguy hiểm...
- Mô phỏng hoạt động với dải rộng trị số linh kiện.

Biểu diễn mạch trong PSpice

- Nguyên tắc: thông qua 2 yếu tố
 - Nút:
 - Là điểm chung của ít nhất 2 phần tử mạch.
 - Mỗi nút được đặt tên duy nhất trong mạch.
 - Mỗi mạch có 1 nút gốc điện áp = 0.
 - Phần tử mạch:
 - Gồm phần tử tích cực và thụ động.
 - Được đặt tên duy nhất trong mạch.
 - Xác định vị trí bởi các nút.
 - Có thông số xác định.

Biểu diễn mạch trong PSpice

- Ví dụ:
 - 2 nút: 0 và 1.
 - 3 phần tử:
 - Nguồn $e(t)$ nằm giữa nút 1 và nút 0.
 - Điện trở R1 nằm giữa nút 1 và nút 0.
 - Điện trở R2 nằm giữa nút 1 và nút 0.



Các file vào ra

- File vào (Input File): dữ liệu đầu vào
 - Lưu thông tin về sơ đồ mạch và chế độ phân tích.
 - Thường là các file *.CIR, *.LIB, *.INC...
- File ra (Output File): kết quả đầu ra
 - Giá trị các tín hiệu đầu ra.
 - Thông báo lỗi trong quá trình mô phỏng.
 - Dữ liệu phục vụ vẽ đồ thị tín hiệu.
 - File *.OUT, *.DAT...

Cấu trúc file vào

- Các thành phần chính

- Dòng tiêu đề nêu tên mạch: optional
- Dòng thành phần: mô tả sơ đồ mạch
- Dòng điều khiển: thiết lập tham số và chế độ phân tích
- .END: kết thúc file vào

```
<Dòng tiêu đề>  
<Dòng thành phần>  
<Dòng thành phần>  
...  
<Dòng thành phần>  
<Dòng điều khiển>  
...  
<Dòng điều khiển>  
.END
```

Biểu diễn giá trị số

- Gồm số nguyên, số thực dấu phẩy động
- Có thể đi kèm đơn vị thứ nguyên hoặc tự xác định theo ngữ cảnh
- Các đơn vị dẫn xuất

Giá trị	Ký hiệu	Giá trị	Ký hiệu
10^{12}	"T" or "t" tera	10^{-3}	"M" or "m" milli
10^9	"G" or "g" giga	10^{-6}	"U" or "u" micro
10^6	"MEG" or "meg" mega	10^{-9}	"N" or "n" nano
10^3	"K" or "k" kilo	10^{-12}	"P" or "p" pico

Một số quy tắc

- Không phân biệt chữ hoa / thường
- Dòng tiêu đề nếu không có thì phải để trống.
- Ý nghĩa các dòng xác định qua ký tự số 0
 - Ký tự '*' : dòng chú thích, được bỏ qua.
 - Ký tự '+' : nối tiếp dòng trước.
 - Ký tự '.' : dòng điều khiển.
 - Ký tự khác: dòng thành phần.

Danh sách các loại phần tử

Ký tự đầu	Mô tả
B	GaAsFET
C	Capacitor
D	Diode
E	Voltage Controlled Voltage Source
F	Current Controlled Current Source
G	Voltage Controlled Current Source
H	Current Controlled Current Source
I	Independent Current Source
J	Junction FET
K	Inductor Coupling (Transformer Core)
L	Inductor
M	MOSFET
Q	Bipolar Transistor
R	Resistor
S	Voltage Controlled Switch
T	Transmission Line
V	Independent Voltage Source
W	Current Controlled Switch

Khai báo điện trở

R<Tên điện trở> <Nút (+)> <Nút (-)> [<Mô hình>] <Giá trị>

- Trong đó
 - Tên điện trở bắt đầu bằng chữ R.
 - Chiều dòng điện trên điện trở quy ước chạy từ nút (+) sang nút (-).
 - Điện áp trên điện trở quy ước lấy theo chiều từ nút (+) sang nút (-).
 - Giá trị được mặc định lấy theo đơn vị Ohm.

Khai báo điện trở

- Ví dụ

Rabc	0	31	14k
rshnt	12	15	99m

- Chú ý

- Nếu giá trị điện trở là số âm thì điện trở được coi là nguồn năng lượng.
- Nếu giá trị = 0 sẽ gây lỗi.
- Giá trị dòng điện/điện áp trên điện trở là số âm nếu chiều dòng điện thực sự đi từ nút (-) sang nút (+).

Khai báo tụ điện

C<Tên tụ> <Nút +> <Nút -> [<Mô hình>] <Giá trị> [<ĐK đầu>]



- Trong đó
 - Tên tụ bắt đầu bằng chữ C.
 - Cực tính các cực của tụ được quy định bởi nút + và nút -.
 - Chiều dòng điện đại số trên tụ quy ước chạy từ nút (+) sang nút (-)
 - Nếu không có tham số [<Mô hình>] thì giá trị của tụ được tính theo đơn vị Fara.
 - Điều kiện đầu là điện áp trên tụ tại thời điểm bắt đầu phân tích

Khai báo tụ điện

- Ví dụ:

C1	1	0	1u	
C2	2	1	10u	5V

- Trường hợp nút bị cách ly
 - Là nút chỉ được nối với các tụ điện.
 - Cần nối thêm điện trở giả có giá trị rất lớn với 1 nút không cách ly nào đó.



Khai báo cuộn cảm

L<Tên> <Nút+> <Nút-> [Mô hình] <Giá trị> [<Điều kiện đầu>]

- Trong đó
 - Tên cuộn cảm bắt đầu bằng chữ L.
 - Chiều dòng điện và điện áp trên cuộn cảm được quy ước chạy từ nút + tới nút -.
 - [Mô hình] là tham số mô hình của cuộn cảm. Nếu không có giá trị tham số mô hình thì giá trị của cuộn cảm được đo bằng Henry.
 - Điều kiện đầu là giá trị dòng điện chạy trong cuộn cảm tại thời điểm bắt đầu phân tích.

Khai báo cuộn cảm

- Ví dụ:

L1 1 2 40mH
L2 3 4 10mH

- Trường hợp các cuộn cảm mắc song song
 - Tương đương các nguồn dòng bị nối song song.
 - Cần nối tiếp với cuộn cảm một điện trở rất nhỏ

Khai báo nguồn

- Nguồn áp một chiều

V|name| |+ node| |- node| {DC} |value|

- Nguồn áp xoay chiều

V|name| |+ node| |- node| AC |magnitude| {phase}

- Nguồn áp đặc biệt


V|name| |+ node| |- node| [transient specification]

Khai báo nguồn

- Trong đó
 - Nguồn dòng bắt đầu bởi ký tự I, nguồn áp bắt đầu bởi ký tự V.
 - Các nút + và nút – là cực tính của nguồn.
 - **value** : giá trị nguồn một chiều.
 - **magnitude** : biên độ nguồn xoay chiều.
 - **phase** : pha ban đầu của nguồn xoay chiều, mặc định bằng 0.
 - **[transient specification]** : chỉ ra kiểu và giá trị nguồn đặc biệt tương ứng.
 - Nguồn xung: PULSE
 - Nguồn mũ: EXP
 - Nguồn điều hòa hình sin: SIN
 - Nguồn tổng hợp: PWL

Các nguồn đặc biệt

- Nguồn xung



- Các tham số



- T_d : thời gian trễ
- T_r : thời gian sườn lên
- T_w : độ rộng xung
- T_f : thời gian sườn xuống
- Period: chu kỳ xung

Nguồn xung

- Khai báo:

PULSE(V1 V2 Td Tr Tf Tw Period)

- Ví dụ:




Vs 1 0 PULSE(0V 10V 0s 100ms 100ms 900ms 2s)

Rs 1 2 10k

Cs 2 0 100uF IC=0V

Nguồn hình sin



Cú pháp


SIN(Vo Va Fr Td Df θ)

Nguồn hình sin

- Trong đó:
 - Vo: điện áp offset DC.
 - Va: biên độ hình sin khi chưa bị suy giảm.
 - fr: tần số dao động.
 - Td: thời gian trễ trước khi dao động.
 - Df: hệ số suy giảm theo thời gian.
 - θ : pha ban đầu.

Nguồn hình sin

- Ví dụ



```
Vs 1 0 SIN(0V 170V 60Hz 0s 0Hz 90d)  
RS 1 2 2k  
Cs 2 0 1uF IC=0V
```


Hàm và tham số

- Khai báo tham số

.param <Tên tham số> = <giá trị>

- Ví dụ: thiết kế mạch phân áp

- Hệ số tỷ lệ 0.2
- Tải đầu vào 50k



```
.param load = 50K
.param ratio = 0.2
Ra 1 2 {load*(1-ratio)}
Rb 2 0 {load*ratio}
```

Các hàm trong PSpice

Tên hàm	Mô tả
ABS(x)	x
SGN(x)	+1 (if x > 0), 0 (if x = 0), -1 (if x < 0)
SQRT(x)	$x^{1/2}$
EXP(x)	e^x
LOG(x)	ln(x) (log base e)
LOG10(x)	log(x) (log base 10)
M(x)	magnitude of x
P(x)	phase of x (result of degrees)
R(x)	real part of x
IMG(x)	imaginary part of x
G(x)	group delay of x (result in seconds)
PWR(x,y)	x^y
SIN(x)	sin(x) (x in radians)

Tên hàm	Mô tả
COS(x)	cos(x) (x in radians)
TAN(x)	tan(x) (x in radians)
ATAN(x)	tan ⁻¹ (x) (result in radians)
ARCTAN(x)	tan ⁻¹ (x) (result in radians)
d(x)	derivative of x with respect to the X axis variable.
s(x)	integral of x over the range of the X axis variable.
AVG(x)	running average of x over the range of the X axis variable.
AVGX(x,d)	running average of X (from x-d to x) over the range of the X axis variable.
RMS(x)	running RMS average of x over the range of the X axis variable.
DB(x)	magnitude in decibels of x.
MIN(x)	minimum of the real part of x.
MAX(x)	maximum of the real part of x.

Chế độ phân tích

- Mô tả chế độ thực hiện phân tích các tín hiệu trong mạch
- 3 chế độ cơ bản
 - DC: phân tích với nguồn 1 chiều trong miền thời gian
 - AC: phân tích xoay chiều trong miền tần số
 - TRANS: phân tích quá độ trong miền thời gian

Phân tích DC

- Thực hiện tính toán giá trị điện áp và dòng điện một chiều trong mạch khi có một thông số nào đó thay đổi.

- Phân tích tuyến tính

**.DC [LIN] <Thông số quét> <Giá trị đầu> <Giá trị cuối>
+ <Bước nhảy> [Vòng phân tích con]**

- Phân tích theo thang logarit

**.DC [OCT,DEC] <Thông số quét> <Giá trị đầu> <Giá trị cuối>
+ <Số điểm> [Vòng phân tích con]**

- Phân tích theo danh sách giá trị

.DC <Thông số quét> LIST <Danh sách giá trị> [Vòng phân tích]

Phân tích AC

- Thực hiện tính toán giá trị điện áp và dòng điện trong mạch ở chế độ xoay chiều khi tần số tín hiệu thay đổi.
- Cú pháp
.AC <[LIN]|DEC|OCT> <Số điểm> <Tần số đầu> <Tần số cuối>
 - **LIN**: phân tích với <Số điểm> trên toàn dải tần số.
 - **DEC**: phân tích trên toàn dải tần số với <Số điểm> trong mỗi decade.
 - **OCT**: phân tích trên toàn dải tần số với <Số điểm> trong mỗi octave.

Phân tích quá độ

- Thực hiện phân tích hoạt động của mạch trong chế độ quá độ, có thể sử dụng các điều kiện đầu trên tụ và cuộn cảm.
- Cú pháp:
.TRAN [/OP] <print step value> <final time value> [UIC]
- → phân tích hoạt động của mạch từ 0 đến thời điểm cuối, với bước nhảy biết trước.
- Ví dụ
.TRAN 20n 1u

Hiển thị kết quả

- PRINT: in kết quả ra file *.OUT dưới dạng text

.PRINT [DC,AC,TRAN] <Các tín hiệu đầu ra>

- PROBE: vẽ đồ thị kết quả

.PROBE [các tín hiệu đầu ra]


➔ Dữ liệu được lưu vào file *.DAT trước khi vẽ

Ví dụ 1

Title: Example 1

Vs	1	0	20.0V
Ra	1	2	5.0k
Rb	2	0	4.0k
Rc	3	0	1.0k
Is	3	2	2.0mA

```
.DC Vs 20 20 1
.PRINT DC V(1,2) I(Ra)
.PRINT DC V(2) I(Rb)
.PRINT DC V(3) I(Rc)
.END
```



Ví dụ 2

Title: Phan tich mot chieu

V1 1 0 DC 120

R1 1 2 120


R2 2 0 240

R3 2 0 120

.DC V1 120 120 1

.PRINT DC V(2) I(R3)

.END



**** 03/30/06 14:09:49 ***** Evaluation PSpice (Nov 1999) *****

Title: Phan tich mot chieu

**** DC TRANSFER CURVES TEMPERATURE = 27.000 DEG C ****

V1	V(2)	I(R3)
1.200E+02	4.800E+01	4.000E-01

Ví dụ 3

Title:Second-Order High-Pass Filter

Vin 1 0 AC 10V

Rf 1 2 4.0


CF 2 3 2.0uF

Lf 3 0 127uH

.AC DEC 20 100Hz 1MEG

.PROBE

.END




Ví dụ 4

```

Title: Phan tich xoay chieu
V1      2      0      AC      24V      0
R1      1      0      600
R2      2      1      200
R3      2      3      600
R4      3      0      200
L1      1      3      0.4
.AC     LIN     1      159.2 159.2
.PRINT          AC      I(L1) V(1)  V(3)
.END

```



**** 03/31/06 09:00:59 ***** Evaluation PSpice (Nov 1999) *****

Title: Phan tich xoay chieu

**** AC ANALYSIS TEMPERATURE = 27.000 DEG C

FREQ	I(L1)	V(1)	V(3)
1.592E+02	2.400E-02	1.610E+01	8.653E+00

Ví dụ 5

Natural Response of a parallel RLC circuit

Rp 0 1 1.0


Lp 1 0 8mH IC=20A

Cp 1 0 10mF IC=0V

.TRAN 500us 100ms 0s 500us UIC

.PROBE


.END



Ví dụ 6

Title: Phan tich qua do

```
V1      1      0      DC      100  
R1      1      2      30  
R2      2      0      20  
C1      2      0      50u      IC=100v  
.TRAN          10ns          4ms      UIC  
.PROBE          V(C1)  
.END
```



Ví dụ 6

